/*
Copyright (c) 2009, Young Digital Planet SA
All rights reserved.

* Redistribution and use in source and binary forms, with or without 
  modification, are permitted provided that the following conditions are met:
* Redistributions of source code must retain the above copyright notice, this 
  list of conditions and the following disclaimer.
* Redistributions in binary form must reproduce the above copyright notice, 
  this list of conditions and the following disclaimer in the documentation 
  and/or other materials provided with the distribution.
* Neither the name of the Young Digital Planet SA nor the names of its 
  contributors may be used to endorse or promote products derived from this 
  software without specific prior written permission.

THIS SOFTWARE IS PROVIDED BY THE COPYRIGHT HOLDERS AND CONTRIBUTORS "AS IS" 
AND ANY EXPRESS OR IMPLIED WARRANTIES, INCLUDING, BUT NOT LIMITED TO, THE 
IMPLIED WARRANTIES OF MERCHANTABILITY AND FITNESS FOR A PARTICULAR PURPOSE 
ARE DISCLAIMED. IN NO EVENT SHALL THE COPYRIGHT HOLDER OR CONTRIBUTORS BE 
LIABLE FOR ANY DIRECT, INDIRECT, INCIDENTAL, SPECIAL, EXEMPLARY, OR 
CONSEQUENTIAL DAMAGES (INCLUDING, BUT NOT LIMITED TO, PROCUREMENT OF 
SUBSTITUTE GOODS OR SERVICES; LOSS OF USE, DATA, OR PROFITS; OR BUSINESS 
INTERRUPTION) HOWEVER CAUSED AND ON ANY THEORY OF LIABILITY, WHETHER IN 
CONTRACT, STRICT LIABILITY, OR TORT (INCLUDING NEGLIGENCE OR OTHERWISE) 
ARISING IN ANY WAY OUT OF THE USE OF THIS SOFTWARE, EVEN IF ADVISED OF THE 
POSSIBILITY OF SUCH DAMAGE.
*/
package pl.ydp.nexl.impl
{
	import flash.events.Event;
	import flash.events.EventDispatcher;
	import org.swiftsuspenders.Injector;
	
	import pl.ydp.nexl.INexlLibrary;
	import pl.ydp.nexl.INexlLibraryContainer;
	import pl.ydp.nexl.Nexl;
	import pl.ydp.nexl.events.LibraryEvent;
	import pl.ydp.utils.Version;

	public class GatheringLibrary extends EventDispatcher implements INexlLibrary, INexlLibraryContainer
	{
		internal var m_libs:Array;
		internal var m_parent:INexlLibraryContainer;
		internal var m_injector:Injector;
		
		public function GatheringLibrary() {
			m_libs = [];
			m_injector = new Injector();
		}
		
		public function get version():Version {
			return Nexl.VERSION_ZERO;
		}
		
		/**
		 * @deprecated
		 * @see addLibrary
		 */
		public function addRegistry( reg:INexlLibrary ):void {
			addLibrary(reg);
		}
		
		/**
		 * Add plugin library to this gathering lib. Plugins from new
		 * library will be available in this lib.
		 * @return true if library was succesfully loaded
		 */
		public function addLibrary( reg:INexlLibrary ):Boolean {
			if ( m_libs.indexOf(reg)==-1 ) {
				m_libs.push( reg );
				reg.addEventListener( Event.COMPLETE, libraryLoaded );
				reg.parent = this;
				reg.injector = m_injector;
				dispatchEvent( new LibraryEvent(LibraryEvent.ADDED) );
				return true;
			}
			return false;
		}
		/**
		 * Remove plugin library from this gathering lib. Plugins from
		 * sub-library will be unavailable.
		 * @return true if library was succesfully removed
		 */
		public function removeLibrary( reg:INexlLibrary ):Boolean {
			var ix:int = m_libs.indexOf( reg );
			if ( ix!=-1 ) {
				reg.removeEventListener( Event.COMPLETE, libraryLoaded );
				reg.parent = null;
				m_libs.splice( ix );
				dispatchEvent( new LibraryEvent(LibraryEvent.REMOVED) );
				return true;
			}
			return false;
		}
		
		protected function libraryLoaded(e:Event):void {
			dispatchEvent( new Event(Event.COMPLETE) );
		}

		public function list(description:Object):Array
		{
			var res:Array = [];
			for each( var lib:INexlLibrary in m_libs ) {
				var sub:Array = lib.list(description);
				if ( sub!=null )
					res = res.concat( sub );
			}
			
			return res;
		}
		
		public function get id():String {
			return null;
		}
		
		public function findLibraryById(s:String):INexlLibrary {
			//
			var l:INexlLibrary = null;
			for each( var lib:* in m_libs ) {
				if (l!=null)
					break;
				if (lib.id==s) {
					l=lib;
					break;
				} else {
					if (lib is GatheringLibrary) {
						l=lib.findLibraryById(s);
					}
				}
			}
			return l;
		}
		
		/* INTERFACE pl.ydp.nexl.INexlLibrary */
		
		public function get injector():Injector {
			return (m_injector ||= new Injector());
		}
		
		public function set injector(value:Injector):void {
			m_injector = value;
		}
			
		public function get parent():INexlLibraryContainer {
			return m_parent;
		}
		public function set parent(s:INexlLibraryContainer):void{
			m_parent = s;
		}
	}
}